Worksheet *AsympCall.mws*

Overview

As describded in the paper *Asymptotics of the price oscillations of a European call option in a tree model* submitted for publication in *Mathematical Finance*,
this worksheet computes the three first coefficients **C0**, **C1**, and **C2** of the asymptotic expansion in powers of **1/sqrt(n)** of the price **C(n)** of a European call option.
Actually, it is not an asymptotic expansion in the usual sens: as explained in the paper such an expansion does not exist; it is an expansion *with bounded coefficients.*
The chosen model is a Cox, Ross, and Rubinstein model, with a quite general choice of the *up* and *down* factors **u** and **d** : it covers in particular popular models such as the *Cox-Rubinstein* model and the *Tian* model.
The first term **C0** in the asymptotic (or "limit price") is the Black-Scholes price **BS**: this result is usually obtained using the Lindeberg-Feller version of the Central-Limit theorem (but this is not the approach used here).
It turns out that the second term, **C1/sqrt(n)** has the difference **(mu-nu)** as a factor; as a consequence, the convergence of **C(n)** towards **BS** is of order **1/n** in the usual models, where **mu=nu** .
The third term **C2(n)/n** where **C2** is a bounded function explains the oscillating behaviour and the scalloped aspect od the curve of **C(n)** with **n** of same parity; **C2** is actually a continuous function of **kappa**, with **kappa(n)** equal to the fractional part of some analytic function **a(n)**, and is thus bounded.
The principle of the computations below is explained in the paper: one begins with an integral version of the price **C(n)** (*proposition 3*) which is:

coef*(S0*I^q-K*exp(-r*T)*I^p)

and one computes separately the asymptotic of the coefficient **coef** and of the two integrales **I^p** and **I^q** . One uses the truncated Stirling formula for **coef** and an *extended Laplace method* for the integrals (*technical theorem 5*).

```
> restart;
```

I.- *The parameters of the model:* **mu**, **sigma**, **nu**, **S0**, **T**, **K,** and **r** are constants; **kappa** is a bounded function, which is "frozen" into a parameter for the asymptotic computations.

```
> u:=proc(n) option remember;
  1+mu*T/n+sigma*sqrt(T/n)+O(1/n/sqrt(n)) end:
> d:=proc(n) option remember;
  1+nu*T/n-sigma*sqrt(T/n)+O(1/n/sqrt(n)) end:
> p:=proc(n) option remember; (exp(r*T/n)-d(n))/(u(n)-d(n)) end:
> q:=proc(n) option remember; p(n)*u(n)*exp(-r*T/n) end:
> a:=proc(n)(ln(K/S0)-ln(d(n))*n)/(ln(u(n))-ln(d(n)))end:
> k:=proc(n) a(n)+1-kappa end:
```

Now one introduces **kk** , a formal asymptotic expansion of **k(n);** the true values of the coefficients of this expansion will be substituted for the formal ones after the computation of the asymptotic of the integral (part IV). Observe the term **O(1/n/sqrt(n))** . In this way we let Maple make sure that enough terms are taken in the expansion of **k(n)** .

```
> kk:=proc(n)
  n/2+k[-1]*sqrt(n)+k[0]+k[1]/sqrt(n)+k[2]/n+O(1/n/sqrt(n))
  end:eval(kk(n));
```

$$\frac{1}{2}n + k_{-1}\sqrt{n} + k_0 + \frac{k_1}{\sqrt{n}} + \frac{k_2}{n} + \mathrm{O}\left(\frac{1}{n^{3/2}}\right)$$

II.- ***Asymptotic of the coefficient*** **c(n,kappa)** defined by formula (15) of the paper (**s2** is the order two expansion of **n!** given by Stirling's formula, and which could be obtained by the Maple command **asympt(n!,n,2);**):

```
> s2:=proc(n)
> sqrt(2*Pi)*exp((n+1/2)*ln(n)-n)*(1+1/12/n+1/288/n/n+O(1/(n^3)))
  end:
> coef:=proc(n)
  exp((1-n)*ln(2))/sqrt(n)*kk(n)*s2(n)/s2(kk(n))/s2(n-kk(n)) end:
> asymptcoef:=map(simplify,asympt(expand(exp(asympt(expand(ln(coef
  (n))),n))),n));
```

$$asymptcoef := \frac{\sqrt{2}\,\mathbf{e}^{\left(-2\,k_{-1}^{\;2}\right)}}{\sqrt{\pi}} - 2\,\frac{\sqrt{2}\,k_{-1}\,(2\,k_0 - 1)\sqrt{\frac{1}{n}}\,\mathbf{e}^{\left(-2\,k_{-1}^{\;2}\right)}}{\sqrt{\pi}}$$
$$-\frac{1}{12}\,\frac{\sqrt{2}\,(16\,k_{-1}^{\;4} + 48\,k_{-1}\,k_1 + 24\,k_0^{\;2} + 3 - 24\,k_0 - 96\,k_{-1}^{\;2}\,k_0^{\;2} - 24\,k_{-1}^{\;2} + 96\,k_0\,k_{-1}^{\;2})\,\mathbf{e}^{\left(-2\,k_{-1}^{\;2}\right)}}{\sqrt{\pi}\,n}$$
$$+\mathrm{O}\left(\frac{\sqrt{\frac{1}{n}}}{n}\right)$$

III.- ***Asymptotic of the two integrals*****, I^p(n,kappa)**, defined by formulas (16) et (17), and **I^q(n,kappa)** which is its analog obtained by substituting the upper bound **q(n)** of the integral for **p(n)**. One computes first the asymptotic of the integral with a formal upper bound **b** which we replace later on by **p(n)** and **q(n)** in order to obtain **I^p** and **I^q**.

```
> integrand:=(1/(1-2*Y/sqrt(n)))*exp((n/2)*ln(1-(2*Y)^2/n)+(kk(n)-
  1-n/2)*ln(((1+2*Y/sqrt(n))/(1-2*Y/sqrt(n))))):
> asymptintegrand:=map(simplify,expand(asympt(integrand,n,3))):
> integralb:=int(eval(subs(O=0,asymptintegrand)),Y=-infinity..b):
> bornep:=map(simplify,eval(subs(O=0,asympt((p(n)-1/2)*sqrt(n),n,3
  )))):
> borneq:=map(simplify,eval(subs(O=0,asympt((q(n)-1/2)*sqrt(n),n,3
  )))):
> integralp:=map(simplify,asympt(subs(b=bornep,integralb),n,2)):
> integralq:=map(simplify,asympt(subs(b=borneq,integralb),n,2)):
```

IV.- ***Computation of the actual coefficients of*** **k(n)** and substitution of these values for the formal ones in the expansions **asympcoef**, **integralp**, and **integralq** of **coef** and of the two integrals. One transforms the type of the expansion of **k(n)** into the type **series** in order to be able to extract its coefficients using the command **op**.

```
> asymptk:=map(simplify,series(eval(subs(n=1/epsilon^2,map(simplif
  y,asympt(k(n),n,3)))),epsilon,2)):k[-1]:=op(3,asymptk);k[0]:=op(
  5,asymptk);k[1]:=op(7,asymptk);k[2]:=op(9,asymptk);
```

$$k_{-1} := \frac{1}{4}\,\frac{2\ln\left(\frac{K}{S0}\right) - T\nu + T\sigma^2 - T\mu}{\sigma\sqrt{T}}$$

$$k_0 := \frac{1}{8}\,\frac{-T\nu\sigma^2 + T\mu\sigma^2 - 2\ln\left(\frac{K}{S0}\right)\mu + 2\ln\left(\frac{K}{S0}\right)\nu - T\nu^2 + T\mu^2 - 8\kappa\sigma^2 + 8\sigma^2}{\sigma^2}$$

$$k_1 := \frac{1}{48}\sqrt{T}\left(-3T\mu^3 - 3T\nu^3 + 2\sigma^6 T + 12\sigma^2\ln\left(\frac{K}{S0}\right)\mu - 12\ln\left(\frac{K}{S0}\right)\mu\nu + 12\sigma^2\ln\left(\frac{K}{S0}\right)\nu\right.$$
$$- 6T\nu\sigma^2\mu - 3T\nu^2\sigma^2 - 2T\nu\sigma^4 - 3T\mu^2\sigma^2 - 2T\mu\sigma^4 + 3T\nu^2\mu + 3T\mu^2\nu$$
$$\left. - 8\sigma^4\ln\left(\frac{K}{S0}\right) + 6\ln\left(\frac{K}{S0}\right)\mu^2 + 6\ln\left(\frac{K}{S0}\right)\nu^2\right) \Big/ \sigma^3$$

$$k_2 := \mathrm{O}(1)$$

```
> coef:=eval(asymptcoef):
> intp:=eval(integralp):
> intq:=eval(integralq):
```

V.- ***Computation of the asymptotic of the price*** **C(n)=coef*(S0*I^q-K*exp(-r*T)*I^p)** and extraction of its three first coeffients **C0**, **C1**, et **C2** using the **<u>series</u>** trick.

```
> AsymptCall:=map(simplify,series(eval(subs(n=1/epsilon^2,coef*(S0*intq-K*exp(-r*T)*intp))),epsilon,3)):C0:=op(1,AsymptCall);C1:=simplify(expand(op(3,AsymptCall)));C2:=simplify(expand(op(5,AsymptCall)));
```

$$C0 := \frac{1}{2}S0\,\mathrm{erf}\left(\frac{1}{4}\,\frac{\sqrt{2}\left(T\sigma^2 + 2Tr - 2\ln\left(\frac{K}{S0}\right)\right)}{\sqrt{T}\,\sigma}\right) + \frac{1}{2}S0$$
$$+ \frac{1}{2}K\,\mathbf{e}^{(-Tr)}\,\mathrm{erf}\left(\frac{1}{4}\,\frac{\sqrt{2}\left(-2Tr + 2\ln\left(\frac{K}{S0}\right) + T\sigma^2\right)}{\sqrt{T}\,\sigma}\right) - \frac{1}{2}K\,\mathbf{e}^{(-Tr)}$$

$$C1 := -\frac{1}{4}\sqrt{2}\,K^{\left(1/2\,\frac{2\ln(S0) + T\sigma^2 + 2Tr}{T\sigma^2}\right)}\,S0^{\left(1/2\,\frac{\sigma^2 - 2r}{\sigma^2}\right)}\,T(-\mu + \nu)$$
$$\mathbf{e}^{\left(-1/8\,\frac{4\ln(K)^2 + 4\ln(S0)^2 + T^2\sigma^4 + 4T^2r\sigma^2 + 4T^2r^2}{T\sigma^2}\right)} \Big/ \sqrt{\pi}$$

$$C2 := -\frac{1}{192}\sqrt{2}\,K^{\left(1/2\,\frac{2\ln(S0) + T\sigma^2 + 2Tr}{T\sigma^2}\right)}\,S0^{\left(1/2\,\frac{\sigma^2 - 2r}{\sigma^2}\right)}\left(24\sigma^4T^2\mu + 24T^2r^2\sigma^2 + 12T^2\nu^2\sigma^2\right.$$
$$+ 192\sigma^4T\kappa^2 - 12T\mu^2\ln(K)^2 + 12T^2\mu^2\sigma^2 - 192\sigma^4T\kappa - 12T\nu^2\ln(S0)^2$$
$$- 16\ln(K)\ln(S0)\sigma^2 - 12T^3r^2\mu^2 + 32\sigma^4T^2r + 16\sigma^4T\ln(K) - 12T\mu^2\ln(S0)^2$$
$$+ 3\sigma^4T^3\nu^2 - 12T^3r^2\nu^2 - 12T\nu^2\ln(K)^2 + 24\sigma^4T^2\nu + 3\sigma^4T^3\mu^2 - 16\sigma^4T\ln(S0)$$

$$- 16\,T\,\mu\,\ln(K)\,\sigma^2 + 16\,T\,\nu\,\ln(S0)\,\sigma^2 - 16\,T\,\nu\,\ln(K)\,\sigma^2 - 48\,T\,\mu\,\nu\,\ln(K)\,\ln(S0)$$
$$- 6\,\sigma^4\,T^3\,\nu\,\mu - 24\,T^2\,r\,\nu^2\,\ln(S0) - 32\,T^2\,r\,\mu\,\sigma^2 + 24\,T\,\mu^2\,\ln(K)\,\ln(S0) + 48\,T^2\,r\,\nu\,\mu\,\ln(S0)$$
$$+ 24\,T\,\mu\,\nu\,\ln(S0)^2 - 48\,T^2\,r\,\nu\,\mu\,\ln(K) + 24\,T\,\mu\,\nu\,\ln(K)^2 + 24\,T^2\,\nu\,\mu\,\sigma^2 + 8\,\ln(K)^2\,\sigma^2$$
$$- 18\,\sigma^6\,T^2 + 8\,\ln(S0)^2\,\sigma^2 - 32\,T^2\,r\,\nu\,\sigma^2 + 16\,T\,r\,\ln(K)\,\sigma^2 - 16\,T\,r\,\ln(S0)\,\sigma^2$$
$$+ 24\,T\,\nu^2\,\ln(K)\,\ln(S0) + 24\,T^2\,r\,\nu^2\,\ln(K) + 24\,T^3\,r^2\,\nu\,\mu + 16\,T\,\mu\,\ln(S0)\,\sigma^2 + 24\,T\,\sigma^4$$
$$+ 24\,T^2\,r\,\mu^2\,\ln(K) - 24\,T^2\,r\,\mu^2\,\ln(S0))\,\mathbf{e}^{\left(-1/8\,\frac{4\,\ln(K)^2 + 4\,\ln(S0)^2 + T^2\,\sigma^4 + 4\,T^2\,r\,\sigma^2 + 4\,T^2\,r^2}{T\,\sigma^2}\right)} \Big/ \left(\sqrt{\pi}\,\sigma^3\right.$$
$$\left.\sqrt{T}\,\right)$$

VI.- ***Plotting***, with a choice of the values for the constants, of the CRR price and its order 0 (**BS** price, in *black*), 1 (in *green*) and 2 (in *red*) approximations. This last one depends on **kappa** and thus oscillates. One observes on the plot how it is difficult to distinguish between the order 2 approximation and the exact CRR price.

```
> n0:=10:n1:=150:K:=1.1:r:=0.05:sigma:=0.2:mu:=0.02:nu:=0.01:S0:=1
  :T:=1:
> call:=proc(n);
> sum(binomial(n,j)*p(n)^j* (1-p(n))^(n-j)
  *'max(S0*u(n)^j*d(n)^(n-j)-K,0)',j=0..n)*exp(-r*T) end:
> approxcall0:=proc(n) evalf(C0) end:
> approxcall1:=proc(n) evalf(C0)+evalf(C1)/sqrt(n)end:
> approxcall2:=proc(n) evalf(C0)+evalf(C1)/sqrt(n)+evalf(C2)/n
  end:
> fraca:=proc(n) frac(a(n)) end:
> with(plots):
> Cexact:=plot(evalf(['[n,call(n)]'
  $n=n0..n1+1]),style=line,color=blue):
> approxC0:=plot(evalf(['[n,approxcall0(n)]'
  $n=n0..n1+1]),style=line,color=black):
> approxC1:=plot(evalf(['[n,approxcall1(n)]'
  $n=n0..n1+1]),style=line,color=green):
> approxC2:=plot(['[n,eval(subs(kappa=fraca(n),approxcall2(n)))]'$
  n=n0..n1],style=point,color=red):
> display({Cexact,approxC0,approxC1,approxC2});
```


0.062
0.0615
0.061
0.0605
0.06
0.0595
20
40
60
80
100
120
140

>